Data Compression and Visual Basic



Wally Foulkes

The genesis of this article should be familiar: Wally read somewhere that it just wasn't feasible to do Huffman compression in VB. Well, he rose to the challenge -- as the accompanying Download file proves. 

There are certain things in the life of a programmer that are as sure as death and taxes -- your hard disk will crash and months of work will vanish, the piece of code you spent the most time on will either turn out to be the reason your program won't work or else cut from the spec at the last minute, and the data stored on your hard disk will rise to meet the disk's capacity. I've experienced all of the above. I also think the problem of diminishing disk space is the easiest to solve thanks to data compression.

The first attempts at data compression were done with hardware. I remember when Run Length Encoding (RLE) was all the rage. RLE simply looked at strings of the same character and saved them as the character followed by a number indicating how long the string was. For example, a string of 20 e's would be encoded e20 -- the original 20 bytes being reduced to two.

Modern compression algorithms are a little more complicated than this, but they still take advantage of duplication in the data. Letters like e, t, r, and s occur more frequently than x or z in most text. Database fields are padded out to fill the declared length. As computers got faster and more powerful, it became practical to write programs to compress and expand data.

Data compression is divided into two distinct worlds: lossy and lossless compression. Lossless compression consists of those techniques guaranteed to produce an exact duplicate of the input data stream after a compress/expand cycle. Lossy compression allows some of the original data to be lost during compression and decompression. Speech, music, and images can lose a lot and still be understood or recognized. On the other hand, databases, spreadsheets, and documents would be corrupted by the loss of a single bit of information. [Our brains are far more tolerant of "fuzzy matches" than the accountants are. -- Ed.]

The more we know about the data content of the files and the environment in which the compression will be executed, the better. After all, a compression routine can be written to maximize compression or throughput. More advanced compression algorithms lean heavily on the field of information theory and the notion of entropy: The greater the content, the higher the entropy. In the context of a message, each symbol's entropy is determined by the probability of its occurrence, expressed in bits. 

PRIVATE
 Number of bits = - Log base 2 (probability) 



If the symbol "e" occurs 20 times in a message and its entropy is 1.26, then its arithmetic coding would be 25.2 bits of encoded data in the complete message. To get the entropy of the message as a whole, you add entropies of all the symbols in the message. In his 1952 seminal paper, "A method for the construction of minimum-redundancy codes," D.A. Huffman introduced a new method for data compression that was state-of-the-art until about 1980. Huffman's method can still be found today in commercial products and is frequently used in two-stage compression with Ziv Lempel sliding window compression (LZH). This article explains lossless compression using the statistical model and the Huffman algorithm.

Huffman encoding 
First, the program reads the file and counts the occurrences of input stream symbols. In this step, the program gathers the statistical information needed to build the compression model. Notice that I use the term symbols, not letters -- the algorithm doesn't care whether the input stream is ASCII, EBCDIC, or binary. It just looks at the bits in the bytes so it can encode any type of file.

Next the program constructs a binary tree from the bottom up based on the symbol counts (probabilities). Trees are described by a number of terms borrowed from other disciplines. The root node is the top of the tree. The nodes might or might not have children. A node without children is called a leaf. A node that has a child or children is a parent node. Branches connect nodes. Nodes have weight. For this algorithm, the weight of a parent node is the sum of the weights of its children. To build the tree, the program finds two nodes without parents that have the lowest weights. It creates a parent node for the two nodes and assigns it a weight equal to the sum of the two nodes. The parent node is added to the list of nodes without parents, and the two child nodes are removed from the list. One child node is designated as the path from the parent node when decoding a 0, the other child node the path when decoding 1.

The program repeats this process of finding minimum weight pairs, creating new parents for them and updating the node list until there's only one node left without a parent. This last node is designated as the root of the tree. The tree becomes the model -- or engine -- of the compressor. To illustrate this, assume you have the following list of symbols and weights:

PRIVATE
 List = {A = 20;  B = 7;  C = 6;  D = 5;  E = 3} 



The two symbols with the smallest weights are E and D. They become the children of the newly created parent node I'll call ED. This parent node's weight is the sum of its children, which is 8. The list update rule says to remove E and D from the list and add the parent ED. The list now looks like this (also see Figure 1):




PRIVATE
 List = {A = 20, B = 7, C = 6, ED = 8} 



Now the nodes with the smallest weights are C and B, and they become the children of the newly created parent node we'll call CB. This parent node's weight is the sum of its children, which is 13. C and B are dropped from the list, and the parent CB added. The list now looks like this (also see Figure 2):




PRIVATE
 List = {A = 20, ED = 8, CB = 13} 



The nodes with the smallest weights are now ED and CB, which become the children of the new parent node I'll call EDCB. This parent node's weight is the sum of its children, 21. ED and CB are dropped from the list, and the parent EDCB is added. The list now looks like this (also see Figure 3):




PRIVATE
 List = {A = 20, EDCB = 21} 



Finally, the only remaining nodes in the list are A and EDCB. They become the children of the newly created parent node we'll call AEDCB. This parent node's weight is the sum of its children, 41. A and EDCB are dropped from the list and the parent AEDCB is added. The list now looks like this (also see Figure 4):




PRIVATE
 List = {AEDCB = 41} 



When there's only one node left in the list, that node is the root node. In the code, I defined the following structure:

PRIVATE
 Type Node 

    Weight As Integer      'Node weight 

    Child1 As Integer      'Branch 1 path 

    Child0 As Integer      'Branch 0 path 

 End Type 



I assigned the smaller weight of the pair to Child0 and the other to Child1. I could have swapped the assignments throughout the tree (see Figure 5). 





The symbol codes would change, but the number of bits in the codes would be the same. Notice that leaf nodes have different numbers of code bits. The symbol A has a 1-bit code, while the other symbols have 3-bit codes. The entropy of the symbol A is the highest, so it occurs most frequently in the data. The tree gives it the fewest bits, so it will take up the least space in the compressed data. How do we handle two or more symbols with the same weights? With rules. 

After building the Huffman weight tree, the program writes the tree counts to the output file. This allows the decoder to rebuild the tree when it decompresses the data. The tree data is the overhead in an encoded file. Counts are saved in front of the encoded data.

Next, the program reads the file again and replaces the symbols it reads with the Huffman codes. Note that the encoder outputs a bit stream, not a series of bytes. Because the symbols that occur most frequently are represented by the shortest strings of bits, the resulting output will be shorter than the input string (see Figure 6).





Huffman decoding 
To decode the compressed data, the program first reads the tree counts from the encoded file. It then recreates the binary tree used to compress the file from the bottom up based on the counts. This process is similar to the one used by the encoder to build the tree. Next, the program reads the Huffman codes from the compressed data and converts them into the corresponding symbols. This process involves reading the data bit by bit and using the bits to walk through the tree to a leaf. When the program reaches a leaf, it outputs the symbol represented by that leaf (see Figure 7).





These Huffman algorithms for encoding and decoding are based on what's called an order 0 statistical model. The order indicates whether or not we look at previous symbols. Order 0 doesn't look at previous symbols, while order 1 looks at one previous symbol, order 2 looks at two previous symbols, and so on.

An order 0 model requires a relatively small overhead of around 256 bytes minimum to save the symbol counts. A higher-order model would give better compression, but the cost in overhead makes it prohibitive. An order 1 model would require the program to save the counts for 256 different trees, or 65,536 bytes minimum.

The VB encoder 
Compress.exe, available in the accompanying Download file, is easy to use. You can type in the path directly or use the file menu, and the output path text box will have a default option based on the path and filename of the input path. To accept this option, press Compress. The file is opened for binary access, read, and the encoded output file is opened for binary access write. The LOF function is used to get the size of the input file in bytes. Counting characters and outputting Huffman codes can take an appreciable amount of time for big files. A percentage complete progress bar is set up by dividing the LOF by 100 and adding 1 to get a ProgressStep variable initialized.

The CharacterCounts subroutine is called to load the Counters array of 256 longs and calls ProgressDisplay at appropriate times to update percent complete. The lstCounts list box is loaded with the raw counts. There's a print routine here if the user decides to print the raw counts. I close the input file at this point, but I'll reopen it later. 

ScaleCounts is the next subroutine called. To keep overhead low, I only want to send integer counts to the output file. Since these tree weights are additive, they can become quite large. The program scales them so the weights can fit within integers. There's room for some experimentation here. I really hate magic numbers, but this routine has one. I can't explain dividing the MaxCount by 255, other than to say that it works. I experimented here some, but in the end I left it alone.

This routine also loads the lstWeights list box with scaled weights (counts), making it easy to compare raw and scaled counts. Selecting a count in one list box selects the corresponding count in the other list box. The exception to this is the weight for node 256. Node 256 is the special EndOfStream (EOS) symbol that the decoder uses to find out where to stop decoding Huffman codes. EOS is assigned a weight of 1.

The BuildTree function creates the binary tree and returns the root node in this user-defined array structure: 

PRIVATE
 Private Type TreeNode 

     Weight As Integer 

     SavedWeight As Integer 

     Child1 As Integer 

     Child0 As Integer 

 End Type 



The Weight parameter holds the scaled weight before the tree is built; it's 0 after the tree is built. The SavedWeight parameter saves the scaled weight for OutputCounts routine. The Child1 parameter contains the 1 branch node if it exists. The Child0 parameter holds the 0 branch node if it exists.

PRIVATE
   Dim Nodes(514) As TreeNode 



There are 257 (0-256) symbols, a possible 256 (257-512) nodes, one dummy (513) count, and one (514) unused.

We know that leaf nodes occupy some or all of the first 256 nodes; therefore, the first free node that can be created is node 257. The variable NextFree is initialized to this value. Dummy count node 513 is initialized to &H7FFF to guarantee that no weight will exceed it as the tree is built.

The variables Min1 and Min2 are initialized to the node 513. Following the rules of the algorithm, Min1 will contain the node of the first instance of a non-zero minimum weight. Min2 will contain the next instance of a minimum non-zero weight.

When creating a parent node, the weights of the nodes in Min1 and Min2 will be added together and placed in the Weight variable of the next free node. Min1's node and Min2's node will have their Weight variables saved in their SavedWeight variables, and their Weight variables will be reset to 0. They're then eliminated from the list in this way.

The parent node is added to the list because it now has a non-zero Weight. Child 0 of the parent node is the node in Min1, and the Child 1 of the parent node is the node in Min2.

This process continues until Min1 contains the last free node and Min2 contains the node 513. The For loop exits, NextFree is decremented by one, and NextFree is returned to the caller as the RootNode.

The For loop checks nodes in ascending order. Several nodes may have the same weight, but Min1 will contain the first one encountered and Min2 the next encountered.

As part of the development process, I drew the tree produced from my test file on paper to verify how the tree was constructed, and I used it later to verify the Huffman codes produced in the next routine. If the file to be encoded contains every ASCII character, the RootNode will be node number 512. In the unlikely event that all nodes have the same Weight, Node 0 and Node 1 will have nine-bit codes, and all the rest will have eight-bit codes, resulting in an encoded file that's larger than the input file.

If the user requests a hard copy of the nodes, the PrintNodes Sub is called and passed the RootNode. This routine formats data from non-zero nodes and sends them to the printer, where I employ another user-defined array:

PRIVATE
   Private Type CharCodes 

      Code As Integer 

      CodeBits As Integer 

   End Type 

   Dim Codes(257) As CharCodes 



The Code parameter holds the Huffman code for the symbol. The CodeBits parameter holds the number of bits in the Huffman code.

Binary trees are great for decoding, but not for encoding in my architecture, so this routine recursively walks the tree in pre-order and loads the Codes array with the Huffman code and the number of bits in the code. The initial arguments are CodeSoFar = 0, Bits = 0, and RootNode. These parameters are pushed on the stack each time the routine is recursively called until a leaf node is reached. The CodeSoFar parameter is shifted left one position, and the Bits are incremented by one on each call until the leaf node is reached. This leaf will be Child 0. At this point, the CodeSoFar is saved as the leaf's Code, and the Bits are saved as the leaf's CodeBits. By ORing a 1 into CodeSoFar, we save the Code and CodeBits for Child 1. When the next recursion takes place, you can't go down any further, so the routine pops the values off the top of the stack and backs up to the current node's parent. At this point, it takes the 1 branch instead of the 0 branch. This process continues until all codes have been generated.

The PrintModel subroutine prints information about the compression model. I used the other print options for debugging, but this option gives the big picture. PrintModel displays a node number, a possible descriptor, SavedWeight, and the code stream. For internal nodes, the routine gives node number, SavedWeight, and the Child nodes. For descriptors, PrintModel calls the PrintCharacter subroutine. The descriptor will be the printable ASCII character, NULL, LF, CR, Space, EOS, or nothing. PrintModel also calls Hex2Bin to display the code stream. The OutputCounts subroutine reduces the overhead in the encoded file by compressing runs of zero counts. The scheme for saving weights is to output FirstNode, LastNode, and the intervening Node().SavedWeights. Outputting a single weight takes three integers.

It searches the first 256 nodes of the nodes array in ascending order, looking for a node with a non-zero SavedWeight and assigns that node to FirstNode. The search continues, but now it looks for a node with a zero SavedWeight and assigns that node to LastNode -1. NextNode is loaded with the node in LastNode +1, and the search continues looking for more nodes with zero weights. If there are more than three zero SavedWeights in a row, FirstNode, LastNode, and the intervening SavedWeights are sent to the output file. FirstNode is then loaded with NextNode +1 and continues as before. If there were less than four zero SavedWeights in a row, LastNode would be loaded with NextNode, and LastNode would continue looking for a node with a zero SavedWeight. The program uses the EndWeightStream (EWS) constant of &HFFF to mark the end of the saved weights.

At this point, the program opens the File to be encoded again and calls the CompressFile subroutine. This routine initializes the Bitio.Mask variable to &H80, reads each byte in the file again, and passes the Huffman code and number of bits in the code to BitHandler. When the input file reaches EOF, the program passes the EOS code and number of bits to BitHandler. It's responsible for updating the percent complete display and seeing that the Bitio.Rack is flushed to the encoded file.

BitHandler is the subroutine that saves the Huffman codes in the encoded file. The Huffman algorithm reads symbols and outputs bits one at a time, but VB can't save data one bit at a time. BitHandler uses the following structure to get around this problem:

PRIVATE
   Private Type BitFile 

      Mask As Byte  

      Rack As Byte 

      OutputCount As Long 

   End Type 

   Dim Bitio As BitFile 



The program loads the "Rack" with the code bits. When the Rack is full, the program writes it into the output file. The Bitio.Mask value points to the bit of interest in the Bitio.Rack. The value iBitCount is used to left-shift the SymbolMask, which points to the bit of interest in the iSymbol. If the SymbolMask is ANDed with iSymbol, and if the result is non-zero, then the Bitio.Mask is ORed with the Bitio.Rack to place a 1 in the Bitio.Rack. This process continues until the SymbolMask is shifted right to 0. If the Bitio.Mask gets shifted right to 0 before the SymbolMask gets to 0, Bitio.Rack is saved to the encoded file, Bitio.Mask is reset to &H80, and the Bitio.Rack is reset to 0. The routine returns to CompressFile to get a new code to save. The program calculates compression percentage using this formula:

PRIVATE
 100 - ((encoded length / input file length) * 100 ) 



A negative compression percentage indicates that the encoded file is bigger than the input file by the negative percentage.

The VB decoder 
To use expand.exe, you identify the file to decompress and the output path, as you did with compress.exe. Fortunately, expansion is a good deal simpler than compression. The VB decoder begins by opening the encoded input file and the expanded output file for binary access.

The InputCounts function uses a Nodes array like the encoder does. It reads the FirstNode integer and the LastNode integer and uses a For loop to load the count integers into the Nodes array. This For loop is inside an endless Do loop, which exits when FirstNode contains EndWeightStream (&H7FFF). The function returns an integer containing the bytes read. The returned integer is used to determine the ProgressStep for the ProgressDisplay. The BuildTree function is exactly like the one used by the encoder. The trees in the encoder and decoder must be exactly alike or the expanded data will be unintelligible.

The ExpandData routine sets the Bitio.Mask value to &H80 and starts an endless Do loop that exits when the program reads EOS. A nested Do loop requests a bit from InputBit routine. The program uses the bits received to walk through the tree to a leaf. Because all leaf nodes are all less than EOS (256), the loop exits and the leaf symbol is output to the expansion file. This routine also calls ProgressDisplay to indicate expansion progress.

The InputBit function reads a byte from the compressed file and loads the Bitio.Rack. The value Bitio.Mask is ANDed with Bitio.Rack, and the result is placed in the Value variable. This variable is non-zero if the mask and rack contain a 1 in the bit of interest. The program then shifts Bitio.Mask right one position. The function returns True if the rack contains a 1 or False if the rack is 0. If the Bitio.Mask = &H00, the mask is reset to &H80. The ProgressDisplay routine works exactly like the one in the encoder.

Summary 
The average compression produced by compress.exe is about 30 percent, which is in line with the figures I've seen for programs written in C. I've used it to compress text files, Word files, and Access files, but it should also work with binary, spreadsheet, and graphics files. The worst compression I got was with two Access databases greater than 1M in size. Huffman compression isn't a speed burner, because it reads the input file twice and does multiple file accesses for a single byte of data. Understanding, not speed, was my main concern when I wrote the programs. You can probably make them a lot faster by changing the way the programs buffer input and output. Error handling is the other obvious area where there's room for improvement. I've listed some resources for compression algorithms in the "Resources" sidebar. Have fun compressing and decompressing. 

Download HUFF.ZIP 

Wally Foulkes is a retired FAA software manager who worked on the program used to control air traffic and the diagnostic software for an IBM 3083 computer. He has also written COBOL report programs for a Tandem computer -- programs that were eventually rewritten in Basic! Wfoulkes@email.msn.com.


Sidebar: Dictionary Pair Compression 

Rod Stephens

Algorithms such as the Huffman compression technique use patterns in the data to compress it. The most complicated parts of the Huffman algorithm are involved in identifying the most frequently used patterns in the data and in converting them into suitable codes.

Sometimes if you know something special about the data you're compressing, you can make this a little easier. For example, if you know the data represents English text, you automatically know a lot about it. You know that the five vowels -- a, e, i, o, and u -- occur fairly often; you know that r and s appear more often than q and z; you know that q is generally followed by u.

You also know that certain letter pairs are more common than others. You can take advantage of this fact to write a relatively simple compression algorithm.

Suppose the text to be compressed contains only printable ASCII characters. These characters lie between <space> (ASCII 32) and ~ (ASCII 127). These 96 characters don't use up all of the 256 possible values you can store in one byte. There are 160 other values that are unused.

The idea behind this encryption technique is to build a dictionary of 160 common letter pairs. As the program scans the message to compress, it tries to find pairs of letters in the dictionary. If it succeeds, the program stores the index of the pair in the compressed message. If the pair isn't in the dictionary, the program stores the first letter in the compressed message. The program continues processing the text until it has all been compressed.

If the program simply stored a character's ASCII code or a pairs index in the dictionary, there would be some overlap. To prevent that, the program maps these values into different values within the range 2 to 255, as shown in Figure 1a. ASCII codes are mapped to the beginning of the range, and pair codes are mapped to the end.





To decompress the data, the program examines a byte value. If the value is 95 or smaller, it represents a character, so the program adds 32 to convert it back into an ASCII code. If the byte value is greater than 95, it's a pair code. The program uses it to look the pair up in the dictionary.

The CompressOrDecompress function shown in Listing 1a performs these conversions. It begins by initializing a dictionary string. This string is a concatenated list of 160 common letter pairs. The pairs begin with <space><space>, e<space>, <space>a, and so forth. The particular pairs in this list were chosen to work well with English text. If you change them, you'll probably reduce the effectiveness of the program on English text, though you might be able to improve compression for other kinds of text.

The program follows the method described previously to compress and decompress text. The only new twist is in the first character. The program sets the value of this character to 255 to indicate that text is compressed. Since uncompressed text must lie between <space> and ~, it can't include the ASCII value 255. The code uses this to determine whether the text it's passed needs compression or decompression.

PRIVATE
Listing 1a. The CompressOrDecompress function does the bulk of the work. 

 ' Compress or decompress a text string. 

 Public Function _ 

    CompressOrDecompress(ByVal input_text As String) _ 

    As String 

 ' Dims omitted  

 ' Initialize the dictionary. 

 dictionary = _ 

    "  e  as  tinthouerhet anreesr d " & _ 

    "onn or o i y wo tontyo. neisarte" & _ 

    "ed,  ctiy  bat snd fal pensestve" & _ 

    "ngitu talehaurllcousa  mf dfoof " & _ 

    "siril  hmeg om Icehironsasiossbe" & _ 

    "depe rli Tetel nicho lilprcactut" & _ 

    "Thpaeceachh wige ebuaisursulmawa" & _ 

    "otowtsmploI solyee Cunm rtieno S" & _ 

    "diwhs.rafincademe.irplk  ury Pwo" & _ 

    "acos gams,duayavucColamowe Aoopu" 

 ' If the first character has ASCII value 255, 

 ' this is a compressed string. 

 If Left$(input_text, 1) = Chr$(255) Then 

 ' Remove the first character. 

    input_text = Mid$(input_text, 2) 

    output_text = ""    ' Start with a blank output. 

    For pos = 1 To Len(input_text) ' examine each char 

    ch_value = Asc(Mid$(input_text, pos, 1)) 

    ' See whether the ASCII value is greater than 95. 

    If ch_value > 95 Then     

    ' This is the code in the dictionary for the two  

    ' chars starting at  pos. 2 * (ch_value - 96) + 1. 

       output_text = output_text & _ 

          Mid$(dictionary, (ch_value - 96) * 2 + 1, 2) 

       CompressOrDecompress = input_text 

    Else ' This is the ASCII value of a char minus 32. 

          output_text = output_text & Chr$(ch_value + 32) 

    End If 

    Next pos 

 Else 

 ' This is a normal string. Compress it. Start with  

 ' Chr$(255) to indicate a compressed string. 

 output_text = Chr$(255) 

 pos = 1 

 Do While pos <= Len(input_text) 

    ' Consider the two chars at pos and pos + 1. 

    pair = Mid$(input_text, pos, 2) 

    ' See whether we've passed the end  

    ' of the input string. 

    If Len(pair) < 2 Then ' We have.  

    ' Set dict_pos = 0 to save this char unencoded. 

       dict_pos = 0 

    Else  ' Find this pair in the dictionary. 

       dict_pos = InStr(dictionary, pair) 

       Do While dict_pos > 0 

       'If pair is here, see whether it's at an odd index. 

          If dict_pos Mod 2 = 1 Then Exit Do 

          ' The pair is at an even position. 

          dict_pos = InStr(dict_pos + 1, dictionary,_ 

             pair) 

       Loop 

    End If 

 ' Add the pair's code or the first char to the output. 

 If dict_pos > 0 Then ' The pair is in the dictionary 

    output_text = output_text & _ 

       Chr$((dict_pos - 1) \ 2 + 96) ' Move past pair 

    pos = pos + 2 

    Else ' The pair isn't in the dictionary. 

    ' Add the first character to the output. 

       output_text = output_text & _ 

          Chr$(Asc(pair) - 32) 

       ' Move past the first char in the input text. 

       pos = pos + 1 

 End If 

 Loop ' Do While pos <= Len(input_text) 

 End If ' End if decompressing ... else ... 

 CompressOrDecompress = output_text 

 End Function

The Compress program, shown in Figure 2a and available in the accompanying Download file, demonstrates the CompressOrDecompress function. Enter some text and click the Compress button to make the program compress it. The program displays the length of the original and compressed text, and the percent reduction in size. It also uncompresses the compressed text to verify that the compression and decompression processes work properly. The program reduced the size of the text in Figure 2a by more than 34 percent. 




Download COMPRESS.ZIP 

Rod has written several Visual Basic books, including Ready-to-Run Visual Basic Algorithms, Second Edition, Bug Proofing Visual Basic, and Visual Basic Graphics Programming. Learn more about his books or download more than 300 example programs at www.vb-helper.com. RodStephens@vb-helper.com.


Sidebar: Resources 
· Binstock, Andrew and Rex, John, Practical Algorithms for Programmers, Addison-Wesley, ISBN 0-201-63208-X, 1998 -- This book contains C language code for Huffman and sliding window compression. Not a must-buy. 

· Nelson, Mark and Gailly, Jean-Loup, The Data Compression Book, M&T, Second edition, ISBN 1-55851-434-1, 1996 -- This book was leaned on heavily for my programs. It has examples in C language, but I didn't have much trouble understanding the code. If you're interested in data compression, this book is a must-buy. 

· Stephens, Rod, Visual Basic Algorithms, John Wiley & Sons, Inc., ISBN 0-471-13418-X, 1996 -- I used this book to understand recursion and binary trees. I consider it a must-buy because it contains so much "ready- to-run" VB code. 
Reading Binary Data Files -- Even with Class -- and Playing Shell Games

Bill Shadish 

I recently lost some valuable time on hoax virus warnings (grhhh). Perhaps I can save you some time by simply alerting you to http://ciac.llnl.gov/ciac/CIACHoaxes.html and http://kumite.com/myths/. Also recommended: http://kumite.com/myths/home.htm and http://sassman.net/virus/. 

Hey, guess what, sports (or should I say, VB) fans? I'm happy to note that this column marks the beginning of my third year as your doctor. And surprise! So far, there have been no known deaths resulting from my doctoral scribings. However, our legal experts do tell me that I must mention that there have been some side effects on some of our readers. Known cases involve wheezing, gasping, and, occasionally, light chuckling. However, the virtual fine print say that this is known to affect less than one percent of you, so, as my helpful and esteemed (he's also a noted TV personality) colleague Dr. Nick has been known to say, "This is a good day to . . . operate." 

And on that note, off we go. 

Binary files?

I have a binary file that I need to open. Each record is 80 bytes in length. I need to open this file with VB5 and parse out the data in each record. Can you please suggest some code to get me started on the right track? 

Gary Cassel (gcassel@home.com)
Baltimore, MD 

I'll assume that your binary file is a collection of ANSI characters kept within a file where the actual record format is known. If you're using an ISAM (indexed sequential access method) file system such as BTRIEVE to create and house your binary data, then you might not be able to use the solution I propose below. That's because ISAM structures can keep a variety of information in their tightly coupled index files and might even store variable-length records. I also assume you know the contents of the file, and if you're working with a binary file whose structure is a mystery, well, you've got a different problem. You might, however, be able to use my approach as a tool to partially automate a brute-force trial-and-error approach to deciphering the contents . . . But, for our purposes, let's assume that this is just a single fixed record length structure file where you know what to expect within each record. 

In that case, it's not too hard to get at the data. First, define your record format in a Type'd data structure, and then use the VB Read statement to pull it out. Given the "record layout" as described in the Employee_Record shown, the Read menu item in this month's example will cycle through the records in the binary file, one at a time. The employee's name for each record, in turn, is shown in a Msgbox that's displayed during the loop (see Listing 1). 

Listing 1. Code to read 80-byte records in a binary file. 

  Type Employee_Record

     SSN As String * 9

     FirstName As String * 15

     MiddleName As String * 15

     LastName As String * 15

     Dept As String * 5           ' Dept Key

     Mgr As String * 5            ' Mgr Key

      SecurityLevel As String * 1  ' Numeric Code

      Title As String * 15 

  End Type

  Dim EmpRec As Employee_Record

  Private Sub mnuRead_Click()

     Dim strMsg$

     Dim BinFile As Long    

     strPath$ = getPath()

     BinFile = FreeFile    

     n strPath$ & "\" & "Temp.Bin" _

        For Binary Access Read As #BinFile Len = 80

     Dim lRecCount As Long   

     lRecCount = 1    

     While Not EOF(BinFile)            

        Get #BinFile, (lRecCount * Len(EmpRec) -_

           Len(EmpRec) + 1), EmpRec    

        MsgBox EmpRec.SSN & ", " & EmpRec.FirstName & _

           ", " & EmpRec.LastName        

         lRecCount = lRecCount + 1   

     Wend    

     Close #BinFile    

  End Sub

While there's plenty of file-specific hard-coding going on in this example -- for example, the record length and the fact that the empRec's variables must be known -- there are a couple of techniques I think you'll find useful. Given our assumptions (80-byte record), the Get method that we use to read the data must be positioned at the beginning of each record manually. That is, Get must be set to 1 the first time through, 81 the second time, and so on. Now, this isn't something you'll want to hard-code, especially if there are a number of different file lengths involved. So, the calculation (lRecCount * Len(EmpRec) - Len(EmpRec) + 1) handles this for us by using the size of the Employee_Record (formerly known as the EmpRec) to advance to the next record each time we travel through the loop. Also, we pull in the entire EmpRec in one read, rather than the more tedious approach of reading in each variable at a time, in order. Notice that it's a simple matter to address the individual variables within the EmpRec once the overall record is retrieved as illustrated in the MsgBox statement. 

Read with class

Now, it might be a little more interesting to build a generic Class file to handle your binaries. This is one of those "write it better once up front and make it easier ever after" scenarios. We'll let our class handle Binary or ASCII files, but only the binary portions are built into the example in Listing 2. The properties for the Class (cFile) to handle your files might look something like the Listing. 

Listing 2. Properties for a binary file-reading class. 

  Public Const EMPLOYEE = 1

  Public Const DEPARTMENT = 2

  Public Const DIVISION = 3

  Public Property Get RecSize() As Variant

     RecSize = mRecSize

  End Property

  Public Property Let RecSize(ByVal vNewValue _

     As Variant)

     If IsNumeric(vNewValue) Then

        mRecSize = vNewValue

     Else

        mRecSize = 1

     End If

  End Property

  Public Property Get FileType() As Variant

     FileType = mFileType

  End Property

  Public Property Let FileType(ByVal vNewValue _

     As Variant)

     If IsNumeric(vNewValue) Then mFileType = vNewValue

  End Property

Notice the FileType property, which allows you to have code handling binary, ASCII, or other types of files. Also, the RecSize property will let us feed in the record size for a file exactly one time (rather than having it hard-coded throughout the application code itself). 

Then we'll add the generic routines to handle the normal Open, Get, and Close VB statements for processing a file (see Listing 3). Notice that we use the properties defined in the class to avoid hard-coding them throughout the application. This makes maintenance easier -- you won't go kicking and screaming if someone wants to up the record size to 81, since a change for this only needs to be made in two places now. Also, since the actual calls to manipulate the file are now found in one place, this more easily allows you to move away from a binary file structure in the future, if so desired. The RecPointer property will be used to automate away the runtime record-keeping of determining which record you're on and the calculations that are required to move you the correct number of bytes into the file to read or add new records. You'll find more routines to handle these basic tasks in the accompanying Download File. 

Listing 3. The class methods. 

  Public Sub bClose(Optional bFile)

     Close mFile

  End Sub

  Private Property Get RecPointer() As Variant

     RecPointer = mRecPointer

  End Property

  Private Property Let RecPointer(ByVal vNewValue _

     As Variant)

     mRecPointer = vNewValue

  End Property

Now we'll add functionality to handle things like parsing a record once we've read it. The Parse routine shown in Listing 4 will trim and separate fields within a record, using a definable character, which is # in our example. The mnuReadWithClass_Click routine is called from the menu, to use our cFile class to read and parse a record. Figure 1 shows the resulting parsed record. 

Listing 4. The class extensions for the parsing routine. 

  Private Const PARSECHAR = "#"

  Public Function ParseRecord(iRecType As Integer) As String

     Dim strBuf$

     Select Case iRecType

        Case EMPLOYEE

           strBuf$ = EmpRec.SSN & PARSECHAR & Trim$_

              (EmpRec.FirstName) & PARSECHAR & _

              Trim$(EmpRec.MiddleName) & PARSECHAR & _

              Trim$(EmpRec.LastName) & PARSECHAR & _

              Trim$(EmpRec.Dept) & PARSECHAR & Trim$_

              (EmpRec.Mgr) & PARSECHAR &_

              EmpRec.SecurityLevel & PARSECHAR & _

              Trim$(EmpRec.Title) & PARSECHAR

     End Select

     ParseRecord = strBuf$

  End Function

  Private Sub mnuReadWithClass_Click()

  '

  'This routine displays the employee record, parsed out

  ' with "#" characters between the fields. (The parse 

  'character can as easily be commas or whatever you'd 

  'like. We return the opened file handle into 

  'FileHandle on open. This syntax allows you to (later) 

  'have multiple files open at the same time. For our  

  'example, we just use the currently opened file,

  ' which is the employee info file.

  Dim objFile As New cFile

  Dim lFileHandle As Long

  Dim strBuf$

  objFile.FileType = BINARY_FILE

  objFile.RecSize = 80 

  lFileHandle = objFile.bOpen(READMODE, getPath() _

     & "\Temp.Bin")   

  clearRec EMPLOYEE    

  strBuf$ = objFile.bRead(lFileHandle, EMPLOYEE)

  MsgBox strBuf$, vbOKOnly + vbInformation, "Parsed Record"

  clearRec EMPLOYEE    

  MsgBox objFile.bRead(lFileHandle, EMPLOYEE), _

     vbOKOnly + vbInformation, "Parsed Record"    

  objFile.bClose

  End Sub

Ahhhh. Problem solved -- happiness is being able to read the binary data file. However, I advise you not to stop here but to write it to a SQL database structure -- for example, in Jet, Access, or SQL Server -- especially if the data is static. A DBMS will provide much more flexibility in terms of reporting, manipulating, and moving the data, as well as maintainability. The only exceptions I can imagine that might keep me from saving the original data out to a SQL database is if the data is frequently changed or if the overall size of your application files is critical. 

Shell games

I've made an EXE file with VB5, and I want to use it in Word97. I use the command shell : 

  Sub do_it()

             Shell ("c:\myprogram.exe"), vbNormalFocus

             msgbox "done"

  End Sub

When I activate the Word macro do_it , myprogram.exe will start. The problem is that the statements after the shelled program are executed immediately. It does not wait for the shelled program to finish, even though the users should click on an OK button in the program (so the focus is in the shelled program). Is there any way to get around this?

René (jusho@wxs.nl),
Amsterdam, Netherlands

This is a fairly common problem -- and not only in the world of VBA. The simplest trick is to use a hard-coded semaphore to hold the Word macro up until the VB app runs. In this context, my hard-coded semaphore is going to be nothing more than a small ASCII file that the Word macro and the VB application will be aware of. This file will be used as a flag to signal which step in the process has control at any given point in time. 

Using something along the logical lines of the following pseudo-code would work to solve your problem (see Figure 2). 

1. Start the Word macro. 

2. Create a control file (in a known location). 

3. Start the VB app. 

4. Fall into a While Loop, within the macro, that checks for the control file. 

5. The VB app completes its work and deletes the control file as part of the shutdown process. 

6. The Word macro then continues happily along. 

Our example adds a parameter ("WAIT") to the VB call within the Word macro (see Listing 5). If that parameter is present when starting the VB app, then the Form Load routine will catch that and kick immediately into the suspense mode, awaiting a user OK click to continue on. The Word macro will wait for this file to disappear and then continue on its way. 

Listing 5. The VB and Word macro code to have VB wait. 

  ' VB code

  Private Sub Form_Load()

     If Left$(UCase$(Command$()), 4) = "WAIT" Then

        Call mnuWait_Click

     End If

  End Sub

  Private Sub mnuWait_Click()

     MsgBox "Let's pretend that this  

     If DoesFileExist("c:\test1.tst") Then Kill _

        "c:\test1.tst"

  End Sub

  'Word Macro Code

  Sub RunApp()

  '

  ' RunApp Macro

  ' Macro created 11/02/98 by Bill Shadish

  '

     Dim lControlFile As Long

     lControlFile = FreeFile   

     Open "c:\test1.tst" For Output Access Write As _

        #lControlFile

     Print #lControlFile, "XYZ"

     Close #lControlFile

     Shell "c:\drvb9901.exe WAIT", vbNormalFocus   

     While DoesFileExist("c:\test1.tst")

     'you may wish to add a timeout warning message here

     Wend

     MsgBox "There, that's better!", _

        vbInformation + vbOKOnly

  End Sub

This control file angle is the simplest approach to use. The only real drawback is that you have to create a separate physical file to use as the process controlling flag. As long as your application checks first for a previously created file and removes it when you're done, there are only a few ways that the file approach can fail. You might abort in the middle of running for some as yet unknown reason. This might leave an errant control file floating around where it isn't expected to be, and so on. Again, proper startup and cleanup code, which is completely reusable within other applications, prevents many of the possible problems. 

Notice that the creation/deletion logic can be reversed. You can create the control file from either the Word macro side or the VB application -- use whichever piece is easier to do. Note that you might want to use the DoesFileExist routine from the preceding Binary File answer. 

If you don't like the ASCII file approach, then you can also store similar control information into a database table. Using a database control table in this way also allows you to keep and track additional data, including the starting and stopping times of all involved components of the application. You can also store successful termination codes or error codes that can become part of an overall status log, available for the reading pleasure of your favorite systems administrator. 

How to send me your questions

Send in your questions via the Dr. VB-ish site at http://www.fo.com/survey/dr-vb.htm. At the bottom of this page is a growing list of the very best links that have been discussed in this column so far. 

Download DRVB199 Sample 

Bill Shadish is a principal of Fundamental Objects, Inc., where he works with ActiveX controls and OLE server technology. (Why, he's even been known to do some Lotus Notes work from time to time!) Bill writes for a number of VB trade journals, including Visual Basic Developer, Inside Visual Basic, VBPJ, and Visual Programming++. bills@fo.com. 

Dynamic MDI Forms—A New Approach

Reprinted with permission from Visual Basic Programmer's Journal, Mar 1999, Volume 9, Issue 3, Copyright 1999, Fawcette Technical Publications, Palo Alto, CA, USA. To subscribe, call 1-800-848-5523, 650-833-7100, visit www.vbpj.com, or visit The Development Exchange at www.devx.com.

Three features in VB combine to make new architectures for dynamically extendable application frameworks.

by Dan Appleman 

Alone, the ability to load controls dynamically is a small change to Visual Basic; the ability to create PropertyBags has been overshadowed by hype about VB's new features; and the Implements statement continues to be difficult to understand. But when you combine these three techniques, entire new application architectures suddenly become possible. 

What you need:

Visual Basic 6.0 

Get the Code 

 The DynamicMDI sample application I'll show you how to build incorporates all these features into a working framework. The framework is built on a standard Multiple Document Interface (MDI) created originally using VB's application wizard. You'll see how the user can select different document types when creating new windows, and save and load MDI documents representing the different document types. 

Figure 1. Three Approaches to a Multiple Document Interface (MDI) Application Click here.

PRIVATE
Three Approaches to a Multiple Document Interface (MDI) Application Some programmers think that a typical MDI app containing a single MDIForm object and one or more regular forms isn't flexible enough. While it might seem logical to improve this by placing each MDI child form into individual ActiveX DLL servers, this solution doesn't work. Instead, here are three ways you can create a dynamically configured MDI application using ActiveX controls. 

Figure 1a. Typical MDI Application  :  The MDI represents one of the most common architectures for Windows applications. An MDI application contains a single MDIForm object, and one or more regular forms that have their MDIChild property set to True (see Figure 1a). Each of these MDI child forms typically represents a different type of document, and more than one of each document type can be visible at a given time. For example, the Excel MDI environment can manage spreadsheet, chart, and macro sheets. The application's main menu can change depending on the document type, and the main application can load or save individual documents. 

Based on the many questions I've received on the subject, it has become apparent to me that many VB programmers want to place the MDI child forms into individual ActiveX DLL servers (see Figure 1b). Such an architecture would offer a number of major benefits. You could update individual documents or document handlers without updating the entire application. You could modify the types of documents the main application could support by adding or removing ActiveX server DLLs from the system without modifying the main application. You could modify the types of documents available to people using the program depending on licensing, user attributes, or any other criteria. You could incrementally add new features to an application by simply shipping out a new ActiveX DLL server with support for a new document type. The only difference is that the individual MDI child forms exist in separate ActiveX DLL servers. 

Figure 1b. Desired, But Flawed, MDI Application Many Visual Basic programmers want to place the MDI child forms into individual ActiveX DLL servers. Such an architecture would offer a number of major benefits. You could update individual documents or document handlers without updating the entire application. You could modify the types of documents the main application could support by adding or removing ActiveX server DLLs from the system without modifying the main application. You could modify the types of documents available to people using the program depending on licensing, user attributes, or any other criteria. You could incrementally add new features to an application by simply shipping out a new ActiveX DLL server with support for a new document type. The major flaw of this architecture is simple: It doesn't work.
The major flaw of this architecture is simple: It doesn't work. VB doesn't allow MDI child forms contained inside ActiveX DLLs to load. It reports an MDIForm is unavailable because it expects the MDI-Form to be in the same application as the child forms. 

 So how about standard forms? Is it possible to somehow place a standard form implemented within an ActiveX DLL into a regular MDI child form? Many VB programmers have tried to do so using the SetParent API function, which forces one window to be the parent of another. This approach does work somewhat, but in every case I've heard of, programmers have run into difficulties-regular MDI keyboard processing stops working correctly and sporadic crashes occur. The truth is VB is not designed with this possibility in mind. 

I've always been skeptical of techniques not based on documented VB features, or not clearly supported by the underlying operating system. It's difficult enough to make applications work if you follow all the rules. The risks of breaking the rules are too great to contemplate for any serious professional work. 

Nevertheless, the idea of dynamic configuration of MDI clients is too good to pass up. Fortunately, VB offers two approaches: a somewhat awkward approach compatible with VB5, and a safe and efficient approach that takes advantage of two of VB6's new features. 

A New Use for ActiveX Controls

An awkward, yet fairly reliable, approach is available under VB5. In this design, the main application contains one type of MDI child form used for every type of document. This MDI child form contains a single Internet Explorer (IE) ActiveX control. The control is loaded with ActiveX documents that represent each type of document. While effective, this approach does have serious disadvantages. 

The approach imposes all the IE control's overhead on each document window (not to mention the application as a whole). It uses ActiveX documents, which have not been widely adopted. When a technology is not widely adopted, the level of support available for that technology in the form of books, articles, information, and support from Microsoft is relatively poor. A significant chance also exists that the technology will be completely abandoned one day, placing the long-term viability of any application adopting that technology at risk. 

The existence of an additional layer (the IE control) between your ActiveX document and the application adds complexity to the entire application, especially when it comes to communication between the main application and your client window software. Nevertheless, this approach points the way to a new strategy that does not suffer from any of these disadvantages. VB6 allows you to dynamically load an ActiveX control onto a form. You can use a generic MDI child form that dynamically loads ActiveX controls that take over the entire client area of the window (see Figure 1c). 

Figure 1c. Dynamic MDI Application VB6 allows you to dynamically load an ActiveX control onto a form. You can use a generic MDI child form that dynamically loads ActiveX controls that take over the entire client area of the window. This architecture is safe, efficient, and uses widely supported technology. But an architecture is only a start. As you'll see, this approach has far-reaching ramifications on the design of the application, and ultimately demands the use of VB6 features still new to many VB programmers.

This architecture is safe, efficient, and uses widely supported technology. But an architecture is only a start. As you'll see, this approach has far-reaching ramifications on the design of the application, and ultimately demands the use of VB6 features that are still new to many VB programmers. 

Loading ActiveX Controls Dynamically

Every VB form contains a Controls property that is a collection of the controls present on the form. VB6 allows you to add new controls to the control collection using its Add method: 

Set EmbeddedControl = _

   Me.Controls.Add(ControlName, _

   NameOfControl)

The full programmatic ID (ProgID) of the control is the first parameter to the Add method. The second parameter contains a string that specifies the value of the Name property for the new control. You can use the Name property later to remove the control from the collection. The Add method returns a reference to the Extender object of the new control. The Extender object contains two sets of properties: the control's properties, and those that VB adds to every control (such as the Left, Top, and Name properties). 

Use the Extender object not only to access the properties of the control, but to receive events as well. You should declare this object at the module level of the form: 

Dim WithEvents EmbeddedControl As _

   VBControlExtender

You'll use a slightly different approach to dynamically load VB intrinsic controls such as the standard Edit or List control. (You can find information on dynamically loading intrinsic controls in the VB documentation.) 

The Extender object's ObjectEvent event processes the control's events, as shown here for the DblClick event: 

Private Sub _

   EmbeddedControl_ObjectEvent( _

   Info As EventInfo)

   If Info.Name = "DblClick" Then

      MsgBox _

         "Control was double-clicked"

   End If

End Sub

How would you apply the ability to add controls dynamically to the dynamic MDI architecture? Each MDI child form contains a single dynamically loaded ActiveX control. To keep things simple, the form uses a single module-level VBControlExtender object called "EmbeddedControl." In this example, the control always has the name "Embedded." The MDI child form has a public method called "CreateControl," which takes the control's programmatic ID as a parameter. The top-level MDIForm creates a new MDI child, and the MDIForm passes the MDI child the name of the control that implements the desired document type and passes it to the CreateControl method (see Listing 1). 

PRIVATE
VB6
The Child Does What the Parent Says

PRIVATE
Public Function CreateControl(ControlName As String)

   If Not EmbeddedControl Is Nothing Then

      Controls.Remove "Embedded"

      Set EmbeddedControl = Nothing

   End If

   ' Error will be raised here if control doesn't load

   Set EmbeddedControl = Me.Controls.Add(ControlName, "Embedded")

   EmbeddedControl.Visible = True

   Form_Resize

End Function

Private Sub Form_Resize()

   On Error Resume Next

   EmbeddedControl.Move 0, 0, ScaleWidth, ScaleHeight

End Sub

PRIVATE
Listing 1 The MDIChild form, whose code is shown here, is responsible for loading the ActiveX control the parent MDIForm specifies.

The CreateControl method first destroys any existing ActiveX control on the form. This allows you to change a given MDI child form's ActiveX control. At first glance, you might think this serves no purpose, but in fact it corresponds to the user opening a new document in an existing window. The CreateControl method next creates the control, makes it visible, and forces a form resize event. The method raises an error if the specified control does not exist, and the parent MDIForm handles this error. 

The form resize event resizes the embedded control so it fills the entire client area of the MDIChild form. The On Error statement eliminates errors that occur if the resize event occurs before a control is loaded, or if a control fails to load due to an error. 

Designing Private Controls

Developing custom ActiveX controls to handle each document type is a key part of the architecture explained in this article. ActiveX controls are by nature general- purpose components you can use in multiple applications. This raises two important questions: Are there features you normally implement in an ActiveX control that a control developed for this architecture doesn't require? Can you restrict your ActiveX control so you can't use it outside your MDIForm application? The answer to both questions is, fortunately, yes. 

The solution to the first problem follows from one of the limitations of ActiveX controls that are dynamically loaded into a form. ActiveX controls can exist in two modes: design mode and run mode. A control exists in design mode when you place it on a VB form in the design-time environment. Design-mode support typically differs from runtime mode in several ways. The control usually displays different information in design mode-often the name of the control. Controls usually support property pages accessible only in design mode. The code used to implement properties within a control usually handles design and run modes differently. In some cases, you can only set properties in design mode. 

When you load a control dynamically, it is loaded in run mode. A control designed for this architecture will be loaded only dynamically, so you don't need to implement many features common to controls. You don't have to implement property pages, design-time display, or any design-time-related code. In fact, you don't have to implement properties at all. How can a useful control have no properties? The answer to that question follows from the solution to making a control private. 

Enforcing Privacy

The controls you use with the approach explained in this article are private in the sense that only your MDI application can use them. But this architecture does require that you implement the controls in separate OCX files, which means they must be public from Windows' point of view-in other words, registered publicly in the system registry. Therefore, you can load the controls into other applications. The trick is to disable all functionality when the controls are loaded in other applications and perhaps display a message stating that the control is not supported in that client. One approach is to disable the control by default and require that the client explicitly turn on the control using some mechanism that only your application knows about. 

You could expose a method that the client could call, but methods are visible through the object browser, and the last thing you want is for people to be able to tinker with your control's properties and methods. A better approach is to have your control implement a private interface that only your application and controls know about. This private interface won't appear in the object browser. 

Creating a new ActiveX DLL server that contains a class whose properties and methods describe the interface is the easiest way to create a private interface with VB. Don't add any code to the class besides the property and method declarations. You will never ship the objects in this server, or even create objects with this server. The idea is to define an interface that your control can implement. 

It's important to think carefully about the properties and methods you wish to include in this interface. It is undesirable, and in most cases dangerous, to change the interface after you deploy a component that uses it. Under this architecture, this private interface is the standard mechanism by which the main MDI application identifies and communicates with the private controls compatible with the application. So be sure you pay adequate attention to the interface design ahead of time (see Listing 2). 

PRIVATE
VB6
Design the Interface

PRIVATE
' This method must be called to enable the control

' This is how we prevent other applications from working

' with this control.

Public Sub Enable()

End Sub

' An example of a common property that you can implement

Public Sub SetUser(ByVal UserName As String)

End Sub

' Method to store the properties of the control into a PropertyBag

Public Sub StoreProperties(P As PropertyBag)

End Sub

' Method to load the properties of the control from a PropertyBag

Public Sub LoadProperties(P As PropertyBag)

End Sub

' Used to obtain a list of menu entries for this control

Public Function GetMenuEntry(Index As Integer) As String

End Function

' Called by the container to indicate that a user clicked a menu

Public Sub MenuClicked(MenuName As String)

End Sub

' Program ID of the control to make embedding easy

Public Function ProgID() As String

End Function

PRIVATE
Listing 2 The private interface is implemented by a class named PrivateMDI in a separate ActiveX DLL, which is never distributed. The PrivateMDI class is shown here.

The parent application calls the Enable method to tell the control it should enable itself. Only your main MDI application can access this interface and call the Enable property because you never distribute the private interface DLL. The interface also includes methods to help store and retrieve the document data managed by the control, and to allow the control to display a custom menu. 

After you finish defining your interface, you should first compile it and set the project to binary compatibility with the newly compiled DLL. This ensures that the low-level class and interface identifiers do not change. You can easily find yourself creating con-trols incompatible with one another and your parent application if you allow these identifiers to change. 

What do you do if you discover later that you must make a change to your private interface? Don't. The solution in this case is to create a new class in the private interface project that contains the interface you need. Then modify your MDIForm application to check for both interfaces, supporting older controls on the first interface, and newer controls on the newer interface. But how do the MDIForm and controls use the private interface in the first place? 

Most of the communication between your main MDI application and the control will be through the private interface. First, support the new interface in your control by adding a reference to the PrivateInterface DLL that contains the PrivateMDI class object and using the Implements statement: 

Implements PrivateInterface.PrivateMDI

Figure 2. Privacy, Please. Click here.

 A control can restrict its functionality outside your main MDI application (see Listing 3).

PRIVATE
VB6
Restrict a Control's Functionality

PRIVATE
Private m_Enabled As Boolean

Private Sub PrivateMDI_Enable()

   m_Enabled = True

End Sub

Private Sub UserControl_Show()

   Dim ctl As Object

   If Not m_Enabled Then

      For Each ctl In _

         UserControl.Controls

         ctl.Visible = False

      Next ctl

      lblPrivate.Visible = True

   Else

      lblGreeting.Caption = _

         "Welcome " & m_User

   End If

End Sub

PRIVATE
Listing 3 Controlling access to a private control is accomplished using a private interface and code to restrict the functionality of the control outside the main MDI application.

 The Enable method of the PrivateMDI interface sets the m_Enabled variable to True and can only be called from your MDI application. The control can have a separate Enabled property on its default interface, but it is independent of the one on the private interface. The m_Enabled variable is False and the control's Show event code hides any constituent controls if the control is placed on a form or container that does not call the private inter-face's Enable method. The control then shows a Label control that contains a warning that the control is not enabled in the environment (see Figure 2). 

PRIVATE


Figure 2. Privacy, Please. If you load a private control on a container other than your custom MDI application, it hides all constituent controls and displays a warning message.

You must modify the MDI child form when using a private interface. Add a reference in your main MDI project to the PrivateInterface DLL so it too can reference the PrivateMDI interface. A variable that is declared as the PrivateMDI type lets you access the control through its PrivateMDI interface. In the form's CreateControl method, add this code immediately after loading the control onto the form: 

Set EmbeddedControl = _

   Me.Controls.Add(ControlName, _

   "Embedded")

EmbeddedControl.Visible = True

Set PrivateInt = _

   EmbeddedControl.object

PrivateInt.Enable

Form_Resize

The Object property of the Embed-dedControl object references the actual control you created instead of the extender. You ask the control for a reference to its PrivateMDI interface when you assign the control to the PrivateInt variable (which is defined as a PrivateMDI type in a module- or function-level Dim statement). Add error handling to deal with this situation in a real application because any attempt to load a control that does not support this interface fails at this point. Once you have that reference, you can call methods and access properties of the interface, effectively using a "back door" to the control that is unique to your application. 

Managing Document Types Dynamically

Each control you create corresponds to a document type. Your application needs some method of determining which document types it will support. A simple approach is to store the list of supported document types and associated control programmatic IDs in the system registry or a text file. You can update this information any time you add a new document type to the application. The File | New menu command brings up a list of available documents. The MDI app can call its CreateControl method, passing it the programmatic ID of the selected control when it opens a new MDI child window: 

Set frmD = New frmDocument

frmDocList.Show vbModal

On Error GoTo nocontrol

If frmDocList.SelectedProgID <> "" _

   Then frmD.CreateControl _

   frmDocList.SelectedProgID

End If

The frmD variable is loaded with a new MDI child form. The frmDocList form represents a form that displays a list of document types from which the user can choose. Use the child form's CreateControl method to load the appropriate control. You need error handling to handle the case where a control fails to load for a given document type. 

Loading and Saving Documents

Creating new documents is simple. Loading and saving documents is where things get truly interesting. You can choose to store each document type in a separate file type with its own extension, or to use a single file type for all documents and store information in the file to distinguish between document types. The latter approach is easier than it sounds and is used in the DynamicMDI example, which uses files with the extension ".mdi". 

The key to making document storage work lies in the way property persistence works in VB6. Visual Basic ActiveX controls have always had the ability to save their state into a PropertyBag. VB6 adds the ability to create PropertyBags and convert them to and from byte arrays. 

You might recall that the private interface defined earlier includes its own LoadProperties and SaveProperties methods. You'll soon see why this is important. Meanwhile, the UserControl ReadProperties and WriteProperties events are delegated to the private interface implementation (see Listing 4). Why do you need this unusual approach for persisting properties? Why not rely just on the UserControl events? Be patient-this will become clear shortly. 

PRIVATE
VB6
Delegate Events

PRIVATE
' Load property values from storage

Private Sub UserControl_ReadProperties(PropBag As _

   PropertyBag)

   Call PrivateMDI_LoadProperties(PropBag)

End Sub

' Write property values to storage

Private Sub UserControl_WriteProperties(PropBag As _

   PropertyBag)

   PrivateMDI_StoreProperties PropBag

End Sub

Private Sub PrivateMDI_LoadProperties(P As PropertyBag)

   FirstNumber = P.ReadProperty("FirstNumber", _

      m_def_FirstNumber)

   SecondNumber = P.ReadProperty("SecondNumber", _

      m_def_SecondNumber)

End Sub

Private Function PrivateMDI_ProgID() As String

   PrivateMDI_ProgID = App.Title & "." & UserControl.Name

End Function

Private Sub PrivateMDI_StoreProperties(P As PropertyBag)

   Call P.WriteProperty("FirstNumber", m_FirstNumber, _

      m_def_FirstNumber)

   Call P.WriteProperty("SecondNumber", m_SecondNumber, _

      m_def_SecondNumber)

End Sub

PRIVATE
Listing 4 This is the property persistence for a control that has two Integer properties. The UserControl events are delegated to the interface methods. The ProgID method obtains the programmatic ID of the control so the MDI application can determine the document type for the child form.

The MDI child form has a method called SaveDocument that obtains a byte array that contains the document data stored with the control. The method creates an empty PropertyBag and uses its WriteProperty method to store the control's object. This causes a single property to be written into the PropertyBag containing a property named "Embedded." This property contains a subobject that consists of the control data written by the control's WriteProperties event, which is raised during the execution of the method: 

Public Function SaveDocument() As _

   Variant

   Dim p As New PropertyBag

   p.WriteProperty "EmbeddedControl", _

      EmbeddedControl.object, Nothing

   SaveDocument = p.Contents

End Function

The PropertyBag's Contents property allows you to retrieve the byte array containing the document information. This array is stored directly into a file by the parent application. When you look into a text version of this information as it is stored in a VB form file, it looks something like this: 

Begin FirstMDI.FirstMDIControl Embedded

   FirstNumber      =   1

   SecondNumber   =   2

End

Re-Creating Controls

Loading a document is, unfortunately, more complex. The MDI parent application passes a byte array containing the file data to the LoadDocument function on the MDI child form (see Listing 5). The MDI child creates a PropertyBag and loads it with the data using the Contents property. The function then calls the ReadProperty event and reads the Embedded property into an object that references the PrivateMDI interface (the private interface defined earlier). This object references a newly created control object with all its properties loaded through its internal ReadProperties event. What type of control is it? The same type you originally stored in the file! The definition of control type (and thus document type) was stored in the PropertyBag during the original WriteProperty call. 

PRIVATE
VB6
Load a Document

PRIVATE
Public Function LoadDocument(ctldata As Variant)

   Dim p As New PropertyBag

   Dim pmdi As PrivateMDI

   p.Contents = ctldata

   Set pmdi = p.ReadProperty("EmbeddedControl")

   Set p = New PropertyBag

   ' Store current property set (without ctl GUID)

   pmdi.StoreProperties p

   ' Create a new control with this GUID

   Call CreateControl(pmdi.ProgId)

   ' We've actually created a second control here, 

   ' which will be placed

   Set pmdi = EmbeddedControl.object

   ' Load current property set (without ctl GUID)

   pmdi.LoadProperties p

End Function

PRIVATE
Listing 5 The LoadDocument method creates a control from information stored in a file.


PRIVATE


Figure 3. Store the Contents You can store the contents of PropertyBags in two ways: using the UserControl persistence events and using the private interface. 

 Now, here's a question for you: Can you use this newly created control? The answer is no! Remember, you dynamically load a control onto a form by using the Add method of the Controls collection. This method requires the programmatic ID of the control, but has no provision for adding an existing control object. You must retrieve the programmatic ID of the control and create a new control of the correct type. You can retrieve the programmatic ID of a control using the ProgID method of the control that you loaded from the PropertyBag (the ProgID function is part of the PrivateMDI interface). You transfer information from the existing control into a newly created empty control by creating a new PropertyBag and storing the data from the existing control into the PropertyBag, then reading the data into the new control. In this case, you use the StoreProperties and LoadProperties methods of the private interface for each control. This way, you only transfer the control's internal data, bypassing the mechanism by which a control stores its type information. This also explains why it is necessary to support two different, but related, property storage mechanisms (see Figure 3). The approach is somewhat inefficient, but it does work. 

Menu Management and Other Extensions

Using a private interface allows you to define functionality common to all the controls your application supports. For example, your document can define custom menus. The MDI child form calls the GetMenuEntry function to obtain a list of menu strings (see Listing 6). The MDI parent form obtains this information from the MDI child form and places it on the menu. When a user clicks on a custom menu item, the command is sent to the MDI child form and then to the MenuClicked method of the private interface. The MDI child form notifies the parent form when it is activated so the MDI form can update its menus based on the current document type. You can extend this simple example to build more complex menus. 

PRIVATE
VB6
Define Custom Menus

PRIVATE
Private Function PrivateMDI_GetMenuEntry(Index As _

   Integer) As String

   Select Case Index

      Case 0

         PrivateMDI_GetMenuEntry = "First Control Menu"

      Case 1

         PrivateMDI_GetMenuEntry = "Second Control Menu"

   End Select

End Function

Private Sub PrivateMDI_MenuClicked( _

   MenuName As String)

   MsgBox "Control Menu " & MenuName & " clicked"

End Sub

PRIVATE
Listing 6 You can use the GetMenuEntry and MenuClicked methods of the private interface to allow the private ActiveX controls to define custom menus to the main MDI form.




Figure 4. Send Data to the Control. You can use the private interface to send common data to the control. For example, the SetUser method allows the parent app to define a user name (Reader) and display it in two different document types. 

 You can also use the private interface to send common data to the control. For example, the PrivateMDI interface includes the SetUser method, which allows the parent application to define a user name. For example, the user name can be Reader, and can be displayed in two radically different document types (see Figure 4). 

When you combine the ability to load controls dynamically, the ability to create PropertyBags, and the Implements statement, entire new application architectures become possible. 

Dan Appleman is the president of Desaware Inc., a developer of software components and add-on products for Visual Basic and other tools. He is the best-selling author of Dan Appleman's Visual Basic Programmer's Guide to the Win32 API, as well as a cofounder and editorial director of APress, a computer book publishing company. You can reach Dan by e-mail at dan@desaware.com. 

Recursion: The Good, the Bad, and the Ugly



Rod Stephens

In this month's column, Rod discusses recursion, a powerful technique for breaking large problems into small ones. Like most other powerful techniques, recursion is subject to abuse, so Rod gives some good and bad examples of recursion and even shows how to remove recursion when necessary. 

Recursion occurs when a function or subroutine calls itself. Programmers often break large problems into smaller pieces, some of which call themselves recursively. The recursive subroutine calls break the pieces into even smaller pieces and call the routine again to handle them. The routine calls itself again and again, breaking the problem into smaller and smaller pieces until they reach a manageable size. At that point, the routine solves the small pieces that make up the larger problem. [Raise your hand if your first introduction to recursion was coding the Tower of Hanoi. -- Ed.]

Recursion is a powerful technique, and some recursive routines perform remarkable computations with surprisingly little code. The Hilbert curve routine described near the end of this article draws curves of amazing complexity with just seven lines of code.

To use recursion, you need to think in terms of breaking a problem into pieces that are similar to the original problem but smaller. This is a bit backwards from the way people normally think about breaking up large problems. Usually people think in terms of performing a series of steps one after another in order to accomplish a task. For example, suppose you need to paint a fence. You'd probably start at one end and paint each board in order until you'd reached the other end of the fence.

To recursively paint the fence, you might subdivide the fence into its left and right halves and then recursively paint the two halves. You would then recursively subdivide each half into quarters, each quarter into eighths, and so on until you were considering pieces of fence that contained a single board. At that point, you'd paint the single boards and be finished. I realize this example is silly; it would take you longer to jump back and forth to paint a fence recursively than it would to just start at one side and work across to the other -- you'd be better served borrowing a page from Tom Sawyer's playbook -- but in computer programs, there are many good examples of recursion.

The good 
The eighteenth-century mathematician Euler discovered a recursive method for finding greatest common divisors. The greatest common divisor (GCD) of two numbers is the largest integer that evenly divides the two numbers. For example, GCD(12, 20) = 4 because 4 is the largest integer that divides 12 and 20 evenly.

The fact Euler discovered is:

PRIVATE
 If A divides B evenly, GCD(A, B) = A. 

 Otherwise GCD(A, B) = GCD(B Mod A, A) 



This leads to a naturally recursive algorithm for finding greatest common divisors:

PRIVATE
 Public Function GCD(ByVal A As Long, ByVal B As Long)_ 

    As Long 

 Dim b_mod_a As Long 

    b_mod_a = B Mod A 

    If b_mod_a = 0 Then 

       GCD = A 

    Else 

       GCD = GCD(b_mod_a, A) 

    End If 

 End Function 



To see how this works, you can follow the steps the GCD function takes to calculate a greatest common divisor. Suppose you call this function to calculate GCD(112, 210):

PRIVATE
 GCD(112, 210)  = GCD(210 Mod 112, 112)  = 

 GCD(98, 112)   = GCD(112 Mod 98, 98)    = 

 GCD(14, 98) 



At this point, the function stops because 14 divides 98 evenly 7 times. That means GCD(112, 210) = 14. You can verify this yourself because 210/14 = 15 and 112/14 = 8. The values 15 and 8 have no common divisors, so 14 is the largest number that evenly divides 112 and 210. Figuring out how quickly the GCD function works is a little tricky. The key is to prove that the second parameter B to the recursive GCD calls shrinks by a factor of at least 1/2 every two times the function calls itself.

Assume that B is originally larger than A. If it's not, you can verify that the first call to GCD switches the order of the parameters and calls GCD again with the values of A and B swapped. After that, B is always larger.

The next call to the function calculates GCD(B Mod A, A). If A _ B/2, then the second parameter in this first recursive call to GCD is already less than or equal to B/2. The second parameter has decreased by a factor of at least one half, and that's all we need to show.

Suppose that isn't the case. In other words, A > B/2 in the original call to GCD(A, B). The next call is GCD(B Mod A, A). Because A > B/2, A divides into B once with a remainder of B - A. That means B Mod A = B A, so the first recursive call to GCD is equivalent to GCD(B - A, A).

The second call to the function computes GCD(A Mod (B - A), B - A). Because we're assuming that A > B/2, the value B - A < B/2. Since B - A is the second parameter in this function call, this shows that the second parameter in the second recursive call to GCD is no more than 1/2 times the original value of B.

All this means that the value for B passed into the GCD function shrinks by a factor of at least 1/2 every two times the function calls itself. Since the recursion must stop when the value for B reaches 1, the function can call itself at most 2 * Log2(B) times. That makes this a very fast routine. For example, if B is on the order of one million, the function can call itself about 2 * Log2(B), or 40 times. In practice, the function is usually called far fewer times that this to find the GCD of even large numbers.

The GCD program, shown in Figure 1 , demonstrates the recursive GDC algorithm. In this example, the function called itself only 18 times to determine that GCD(1,289,617,875, 1,994,838,720) = 105.




The bad 
Recursion isn't always the best solution, even when you can figure out how to apply it. For example, Fibonacci numbers have a naturally recursive definition:

PRIVATE
 Fibo(0) = 0 

 Fibo(1) = 1 

 Fibo(N) = Fibo(N - 1) + Fibo(N - 2) for N > 1 



Table 1 shows the values of the first several Fibonacci numbers. The values grow extremely quickly when N is large. For example, Fibo(30) = 832,040.

Table 1 . Fibonacci numbers.
PRIVATE
N
0
1
2
3
4
5
6
7
8
9
10

Fibo(N)
0
1
1
2
3
5
8
13
21
34
55



The recursive definition of the Fibonacci numbers leads to the following simple recursive function:

PRIVATE
 Public Function Fibo(ByVal N As Integer) As Double 

     If N <= 1 Then 

         Fibo = N 

     Else 

         Fibo = Fibo(N - 1) + Fibo(N - 2) 

     End If 

 End Function 



Program Fibo, shown in Figure 2 , demonstrates this algorithm. In addition to calculating Fibonacci numbers, the program displays the number of times the Fibo function was executed. In this example, the function was called more than 2.6 million times to calculate the 30th Fibonacci number. As you might imagine, making all those recursive function calls takes the program a long time.

The reason this routine takes so long is that it recalculates a huge number of intermediate values. To compute Fibo(N), the function calls itself with parameters N - 1 and N - 2. To calculate Fibo(N - 1), the routine must calculate Fibo(N - 2) and Fibo(N - 3). Here the value Fibo(N - 2) is calculated twice.




Figure 3 illustrates the function calls used by the Fibo function to calculate Fibo(6). Each number represents a Fibonacci value. Nodes are connected to indicate the intermediate values they must calculate. For example, the root node 6 is connected to nodes labeled 5 and 4 to indicate that the function must calculate Fibo(5) and Fibo(4) to calculate Fibo(6).




Looking at Figure 3 , you can see all of the repeated calculations. The values Fibo(0) and Fibo(1), for example, occur a total of 13 times. In fact, it can be shown that in calculating Fibo(N), the function calculates Fibo(0) and Fibo(1) a total of Fibo(N + 1) times. You can verify this in Figure 3. To calculate Fibo(6), the program evaluates Fibo(0) and Fibo(1) a total of Fibo(6 + 1) = 13 times.




For larger values of N, Fibo(N) grows extremely quickly. For example, Fibo(31) = 1,346,269. That means to calculate Fibo(30), the program recalculates Fibo(0) and Fibo(1) more than 1.3 million times. It's in calculating and recalculating all of these intermediate values that the program spends all of its time.

This is an example of a bad use for recursion. Recursion lets the Fibo function break its calculation into two smaller problems, but the two problems have a lot of overlap. They don't share the work very well and, in fact, duplicate some of each other's efforts. To prevent this wasted duplication, you need to rethink the definition of the Fibonacci numbers. The recursive function uses a top-down approach. While the input number is greater than 1, the function breaks the problem into two smaller problems and solves them recursively.

An alternative bottom-up strategy starts with the known values for Fibo(0) and Fibo(1). It combines those values to produce Fibo(2). It then combines Fibo(1) and Fibo(2) to calculate Fibo(3). It continues combining smaller values to build bigger values until it reaches the value of Fibo(N) that it needs. Instead of breaking a big problem into small pieces, it starts with small values and combines them to calculate bigger values.

Program Fibo2 uses the following code to compute Fibonacci numbers. Although Fibonacci numbers are integers, this code performs its calculations using doubles so it can compute approximate values for very large Fibonacci numbers.

PRIVATE
 Public Function Fibo(ByVal N As Integer) As Double 

 Dim fibo_i_minus_2 As Double 

 Dim fibo_i_minus_1 As Double 

 Dim fibo_i As Double 

 Dim i As Integer 

    ' Set Fibo(i) and Fibo(i - 1) for i = 1. 

    fibo_i_minus_1 = 0 

    fibo_i = 1 

    ' Calculate larger values for Fibo(i). 

    For i = 2 To N 

       fibo_i_minus_2 = fibo_i_minus_1 

       fibo_i_minus_1 = fibo_i 

       fibo_i = fibo_i_minus_1 + fibo_i_minus_2 

    Next i 

    Fibo = fibo_i 

 End Function 



Figure 4 shows program Fibo2 calculating Fibo(1476), the largest value it can compute without causing a double precision overflow. While program Fibo takes more than 10 seconds to calculate Fibo(30) on a (vintage?) 133 MHz Pentium, program Fibo2 takes almost no time to calculate Fibo(1476).




The ugly 
Sometimes it's useful to rewrite a recursive algorithm in a non-recursive way. Rewriting the Fibo function saves it from a huge amount of duplicated effort. Some programming languages don't allow recursion, so you need to implement all functions non-recursively in those languages. Early versions of VB also had a limited stack. [In fact, all it takes to do recursion is the ability to pass parameters on the stack and create local variables, something almost all "modern" programming languages let you do. -- Ed.] A recursive routine could only call itself a certain number of times before exhausting the stack space and crashing. The exact depth of recursion possible depended on the amount of memory available and on the amount of memory used by each function call.

It's not always easy to rewrite an algorithm non-recursively, but it's always possible. For an example that's more complex than the Fibo function, consider the Hilbert program, the results of which are shown in Figure 5 . The Hilbert subroutine is multiply recursive, calling itself in four different places.




PRIVATE
 Private Sub Hilbert(ByVal depth As Integer, ByVal dx _ 

    As Single, ByVal dy As Single) 

    If depth > 1 Then Hilbert depth - 1, dy, dx 

    picHilbert.Line -Step(dx, dy) 

    If depth > 1 Then Hilbert depth - 1, dx, dy 

    picHilbert.Line -Step(dy, dx) 

    If depth > 1 Then Hilbert depth - 1, dx, dy 

    picHilbert.Line -Step(-dx, -dy) 

    If depth > 1 Then Hilbert depth - 1, -dy, -dx 

 End Sub 



Rewriting this routine from scratch without recursion could be quite difficult because it doesn't have simple top-down and bottom-up versions for you to use. Fortunately, there's a methodical system for converting recursive algorithms into non-recursive ones.

The idea is to simulate the steps the computer takes to perform recursion. Before calling the recursive routine, the computer saves the information it will need to continue running after the recursive call finishes. It saves any variable values that it will need later. It also saves its current location in the code so it knows where to continue execution later. The line of code the computer is currently executing is stored in a program counter, or pc for short.

To mimic these steps in VB, start by labeling all of the places the program will need to resume execution in the routine. This includes the beginning of the routine and each of the lines that follow a recursion. (The labels in the following code are to help you keep track of where the program should continue execution and aren't actually part of the VB code.)

PRIVATE
 Private Sub Hilbert(ByVal depth As Integer, ByVal dx _ 

    As Single, ByVal dy As Single) 

 1  If depth > 1 Then Hilbert depth - 1, dy, dx 

 2  picHilbert.Line -Step(dx, dy) 

    If depth > 1 Then Hilbert depth - 1, dx, dy 

 3  picHilbert.Line -Step(dy, dx) 

    If depth > 1 Then Hilbert depth - 1, dx, dy 

 4  picHilbert.Line -Step(-dx, -dy) 

    If depth > 1 Then Hilbert depth - 1, -dy, -dx 

 End Sub 



Next, create a collection for each value the program needs to save before simulating a recursion. This includes any variables the routine will need after the recursion and the program counter. The saved program counter will tell the program where to continue execution after a recursion finishes.

You can make saving and restoring values easier by writing SaveValues and RestoreValues subroutines. These routines take values to save or restore as parameters and then add or remove them from the collections.

To rewrite the recursive routine, use an integer variable pc to keep track of the code segment that the routine should be executing. Place all the code inside a big While loop that contains a Select statement. The cases in the Select statement should be the labels used to divide the code earlier. Each time it passes through the While loop, the Select statement makes the program execute the code indicated by the pc variable. Listing 1 shows the general structure of the non-recursive Hilbert subroutine.

PRIVATE

Listing 1

. The general code structure for the non-recursive Hilbert subroutine. 

 ' Draw a Hilbert curve. 

 Private Sub Hilbert(ByVal depth As Integer, ByVal dx _ 

    As Single, ByVal dy As Single) 

 Dim pc As Integer 

    ' Start at label 1. 

    pc = 1 

    ' Repeat until we're done. 

    Do 

       ' Execute the correct piece of code. 

       Select Case pc 

          Case 1 

             ' Do code block 1 stuff. 

                : 

          Case 2 

             ' Do code block 2 stuff. 

                : 

          Case 3 

             ' Do code block 3 stuff. 

                : 

          Case 4 

             ' Do code block 4 stuff. 

                : 

          Case 0 

             ' Return from a recursion. 

                 : 

       End Select 

    Loop 

 End Sub 



Now you can finally fill in the details. When the code needs to perform a recursion, save the current variable values and the pc value that tells where the routine should continue after the recursion. For example, after the recursion at label 1 returns, the program should continue execution at label 2.

Next, change the variables so they're ready for the recursive subroutine call. In the recursive call at line 1 in the original code, the routine calls itself with the values dx and dy reversed. To mimic this behavior, the non-recursive routine must swap the values of dx and dy. This is why you save the variable values in collections. When the recursion returns, you can use the collections to restore them to their previous values.

Finally, set the pc variable to 1. The next time the routine enters its Select statement, this makes it execute the first line of code in the routine. That makes the recursive call start execution at the beginning of the routine. The last detail you need to handle is returning from recursion. In the original recursive code, after the routine finishes executing the code beginning with label 4, it returns and execution resumes with the calling instance of the routine. The non-recursive version must mimic this behavior. When the program finishes executing the code beginning at label 4, it sets pc to 0. This indicates that the routine is finishing a recursion. The next time through the While loop, the Select statement executes code for returning from a recursion.

This code first determines whether the value collections are empty. If they are, that means no recursive subroutine call is in progress. In that case, the program is done with the original subroutine call. The routine exits its Do loop and is finished. If the saved value collections aren't empty, the routine restores the most recently saved values. When the routine next enters its Select statement, it continues execution at the line indicated by the newly restored pc value.

The complete VB source code for the non-recursive Hilbert subroutine (Hilbert2) is available in the accompanying Download file . As you'd expect, the result is identical in appearance to the curves drawn by the original recursive version.

In this case, converting a recursive routine into a non-recursive one doesn't make the routine any faster, because I've simply mimicked the steps the computer takes in the recursive version, so it doesn't save any steps. (In fact, VB can probably perform these steps faster than the program can mimic them, so performance might decrease slightly.) In this example, however, the recursive Hilbert subroutine is easier to understand and a little faster, so there's no real need to use the non-recursive version. It just makes an interesting example.

On the other hand, this non-recursive technique prevents deep recursion and gives you direct control over the routine's memory allocation. If you know in advance how much memory the routine will need, you might be able to allocate it all at once instead of making each recursive subroutine call allocate its own memory. That might make the routine a bit faster.

Conclusion 
Recursion is a powerful technique. It allows a relatively simple routine to break a complex problem down into smaller pieces until each piece is manageable. The Hilbert subroutine is a good example because it produces a remarkably complex picture with just a few lines of code.

On the other hand, recursion isn't always the best approach. In some cases, you can use a bottom-up approach to remove recursion from an algorithm. In the Fibonacci example, this approach leads to a routine that's much more efficient than the recursive version. In other cases, you can use a methodical system to remove recursion from any recursive subroutine. This might not produce the most concise code possible, but it does give you greater control over the routine's memory requirements. 

Download ROD1198 Sample. 

You can learn a lot more about recursion and other algorithmic techniques in Rod's book, Ready-to-Run Visual Basic Algorithms, Second Edition. To download more than 200 example programs or to learn more about his books, visit Rod's Web site at www.vb-helper.com . RodStephens@vb-helper.com. 


Send feedback to MSDN. Look here for MSDN Online resources. 

